ElmのTaskにこんにちは

Elmは基本的に作りがシンプルで理解に詰まることが少ない言語ですが、それでもやはりプログラミング言語である以上多少なりとも壁は存在します。そのうちの一つが[Task](https://package.elm-lang.org/packages/elm/core/latest/Task)になります。Taskはいろんな説明ができてしまいますが(~~非同期を扱うモジュール・および型、Effect Managerを間接的に操作するための機構~~)、Elmの場合せっかくThe Elm Architectureと言うわかりやすい機構があるので、Elm Runtimeに処理をお願いできるもの。ぐらいの認識で十分です。詳しく原理がわからないと落ち着かんと言う人は、[基礎からわかるElm](https://www.amazon.co.jp/dp/B07P253FPG)をどうぞ。

# TaskはCmdを作るもの

TaskはいろんなことができるのですがElmの素晴らしくシンプルでわかりやすいこととして、**どれだけTaskをいじくり回しても必ずCmd msgにしなければなりません**。Taskを使う上で避け得られない関数が、[perform](https://package.elm-lang.org/packages/elm/core/latest/Task#perform)と[attempt](https://package.elm-lang.org/packages/elm/core/latest/Task#attempt)の二つです。TaskはElmランタイムつまりJavaScriptに命令を出しますが、Elmは自分以外を信用しない臆病な生き物なので、それは成功するかもしれないし失敗するかもしれないと思って動きます。そんなことを言っても成功するでしょう、もし仮に失敗したとしても失敗を検知しませんよ。と言うのがperformです。わかりました失敗に備えてエラーのハンドリングもしっかりします。と言うのがattemptになります。いずれもmsgが受け取る値が`a`型になります。Elmの型システム上、絶対起きえないと言うことでも明記しておかないと怒られてしまうので、そんな時コンパイラを黙らせる型が[Never](https://package.elm-lang.org/packages/elm/core/latest/Basics#Never)型になります。エラーハンドリングを扱う場合には、[Result](https://package.elm-lang.org/packages/elm/core/latest/Result)型を使います。エラーは処理に依存するので固変数`x`となっています。

```elm
perform : (a -> msg) -> Task Never a -> Cmd msg

attempt : (Result x a -> msg) -> Task x a -> Cmd msg
```

## performでHello World

それでは、TaskでHello Worldをしてみましょう。デモとコードは[こちら](https://ellie-app.com/7psD5DGBMLta1)。これは単なるStringをランタイムから受け取るためのプログラムです。ランタイムから受け取るため基本的に中身はブラックボックスなのですが、固定値であればElm側から指定することができます。必ず成功する固定値を返して欲しい場合には、[Task.suceed](https://package.elm-lang.org/packages/elm/core/latest/Task#succeed)を利用します。これは誰が見ても必ず成功するだろうと言うことで、performを利用して失敗は無視します。

```elm
type alias Model =
    String


init : () -> ( Model, Cmd Msg )
init _ =
    ( ""
    , Task.perform GotGreet <| Task.succeed "Hello World"
    )



-- UPDATE


type Msg
    = GotGreet String


update : Msg -> Model -> ( Model, Cmd Msg )
update msg _ =
    case msg of
        GotGreet greet ->
            ( greet, Cmd.none )
```

performを使って処理すべきTaskを返すものとしては、[Time.now](https://package.elm-lang.org/packages/elm/time/latest/Time#now)・[Time.here](https://package.elm-lang.org/packages/elm/time/latest/Time#here)・[Random.generate](https://package.elm-lang.org/packages/elm/random/latest/Random#generate)・[Process.sleep](https://package.elm-lang.org/packages/elm/core/latest/Process#sleep)があります。sleepはアニメーションなどを扱う場合に重宝するでしょう。

## attemptでHello World

続いて、失敗するかもしれないことを考慮するattemptを利用してみます。デモとコードは[こちら](https://ellie-app.com/7pt3yWDVRVqa1)。succeedを利用している限り必ず成功してしまうため、必ず失敗をする[Task.fail](https://package.elm-lang.org/packages/elm/core/latest/Task#fail)を使います。もし、起こり得るエラーが複数種類ある場合やエラーであることを明示するためにはカスタムタイプでError型を用意して使うのが好ましいです。

```elm
type alias Model =
    { ok : String
    , error : String
    }


type Error
    = SomeError String


init : () -> ( Model, Cmd Msg )
init _ =
    ( { ok = "", error = "" }
    , Task.attempt GotResult <| Task.fail <| SomeError "なんか失敗しました"
    )



-- UPDATE


type Msg
    = GotResult (Result Error String)


update : Msg -> Model -> ( Model, Cmd Msg )
update msg model =
    case msg of
        GotResult result ->
            case result of
                Ok v ->
                    ( { model | ok = v }, Cmd.none )

                Err err ->
                    case err of
                        SomeError ev ->
                            ( { model | error = ev }, Cmd.none )
```

attemptを使って処理すべきTaskを返すものとしては、[Http.get](https://package.elm-lang.org/packages/elm/http/latest/Http#get)(post, request)・[Browser.focus](https://package.elm-lang.org/packages/elm/browser/latest/Browser-Dom#focus)・[Browser.blur](https://package.elm-lang.org/packages/elm/browser/latest/Browser-Dom#blur)などがあります。Httpの場合は、BadRequestやNotFound、InternalServerErrorなどをハンドリングしたり、Browserなどの場合は、要素が見つからなかった(NotFound)時のハンドリングを要求されます。

[ElmでHttpをわかってしまおう](https://qiita.com/ababup1192/items/b03fce202e1018bc4992)と言う別途記事でHttpやTaskの応用的な扱い方を紹介しています。良ければご覧ください。

# まとめ

本当はTask.mapやandThenも紹介しようと思ったのですが、一番の混乱する理由は最終的にCmdに落とされると言う点とperform, attempt何故2種類あるの？どう扱うの？ と言う点で、あとは具体的な関数のリファレンスに使い方が記されているケースがほとんどなので、これ以上深入りはしません。Taskであまり混乱しないようにするポイントはTaskとはなんなのか？と考えすぎない点とTaskの使い方自体はあまり追うのではなく、Taskを返す関数の使い方をしっかり読む、TEAやResult型などの基本をしっかり理解することだと思います。それでは、良いElmライフを！